

UNIVERSIDAD TECNOLÓGICA NACIONAL - FACULTAD REGIONAL ROSARIO
Departamento de Ingeniería Química

Cátedra: Integración IV

Tema: Solución de una Ecuación de Estado Mediante el Método de Newton-Raphson

Alumnos: Damián Matich, Marcos Bossi y Juan M. Pignani

Profesores: Dr. Nicolás Scenna, Dr. Alejandro Santa Cruz y Dra. Sonia Benz

Año de cursado: 1999

Ecuación de Estado:

Escribir un programa que utilice el método de Newton-Raphson para resolver la siguiente ecuación de estado:

$$P = \frac{RT}{V} + \frac{\beta}{V^2} + \frac{Y}{V^3} + \frac{\delta}{V^4} \quad (1)$$

donde:

- P : es la presión, [atm].
- V : es el volumen, [lt/gmol].
- T : es la temperatura, [°K].
- β , Y y δ : son los parámetros característicos del gas dependientes de la temperatura.
- R : es la constante universal de los gases, [(atm lt)/(°K gmol)].

Los parámetros β , Y y δ se definen como:

$$\beta = RT B_0 - A_0 - \frac{Rc}{T^2}$$
$$Y = -RT B_0 b + A_0 a - \frac{Rc B_0}{T^2}$$
$$\delta = \frac{R B_0 b c}{T^2}$$

donde: A_0 , B_0 , a , b , y c son constantes determinadas en forma empírica y que se hallan tabuladas (diferentes para cada gas).

La ecuación (1) es explícita en la presión, pero implícita en la temperatura y el volumen. Por consiguiente, se requiere de un método iterativo para encontrar el volumen que corresponde a valores dados de presión (1, 2, 5, 10, 20, 40, 60, 80, 120, 140, 160, 180 y 200 atm) y temperatura (273 y 473 °K).

Formulación del problema:

A través del método de Newton-Raphson resolver la ecuación de estado para el volumen molar del gas metano (CH₄), dadas la presión, temperatura y constantes R , A_0 , B_0 , a , b y c . A continuación se calcula, con el valor de V hallado, el factor de compresibilidad z . Siendo, $z = \frac{PV}{RT}$, un índice útil de la desviación al comportamiento de un gas ideal ($z = 1$)

Resolución del problema:

Método de Newton-Raphson:

$$y^{n+1} = y^n - \frac{f(y^n)}{f'(y^n)}$$

La Ley de los Gases Ideales suministra una primera aproximación para el volumen molar, $V = \frac{RT}{P}$ y así inicializar el método iterativo.

El criterio de finalización del proceso iterativo es: $\left| \frac{V_{n+1} - V_n}{V_{n+1}} \right| \leq \epsilon$, donde ϵ es un número positivo pequeño (10^{-N}). El método de N-R podría no converger si el valor e arranque es malo, por lo que el número de iteraciones debe limitarse.


Para hallar $V = f(P, T)$ a través de N-R es necesario reescribir la ecuación de estado de la siguiente forma:

$$f(V) = \frac{RT}{V} + \frac{\beta}{V^2} + \frac{Y}{V^3} + \frac{\delta}{V^4} - P = 0$$

Las constantes apropiadas para el metano son: $A_0 = 2.2769$, $B_0 = 0.05587$, $a = 0.01855$, $b = -0.01587$, $c = 12.83 \cdot 10^4$


Resultados:

$T = 273^\circ\text{K}$			$T = 473^\circ\text{K}$		
$\beta = -1.16735$			$\beta = -0.15695$		
$Y = 5.41 \cdot 10^{-2}$			$Y = 7.39 \cdot 10^{-2}$		
$\delta = -1.25 \cdot 10^{-4}$			$\delta = -4.16 \cdot 10^{-2}$		
P (atm)	V (lt/gmol)	z	P (atm)	V (lt/gmol)	z
1	22.333	0.997	1	38.782	0.999
2	11.140	0.995	2	19.389	0.999
5	4.424	0.988	5	7.753	0.999
10	2.186	0.976	10	3.875	0.999
20	1.066	0.953	20	1.936	0.998
40	0.507	0.906	40	0.967	0.997
60	0.321	0.860	60	0.645	0.998
80	0.228	0.817	80	0.484	0.999
120	0.139	0.748	120	0.325	1.005
140	0.116	0.727	140	0.279	1.009
160	0.100	0.715	160	0.246	1.014
180	0.088	0.711	180	0.219	1.020
200	0.080	0.715	200	0.199	1.027



Programación:

El método de Newton-Raphson para calcular los valores del volumen molar y del coeficiente de compresibilidad se programó en QBASIC.



A continuación se muestra el listado del programa:

```

SCREEN 12
DIM Respuesta AS STRING
DO
  PRINT
  PRINT "Ingrese el número de Temperaturas";

```

```

INPUT N
PRINT
PRINT
PRINT "¿El valor es correcto? (S,N)";
INPUT Respuesta
IF (Respuesta = "N") OR (Respuesta = "n") THEN
    Continuar = 0
ELSE
    Continuar = 1
END IF
LOOP UNTIL Continuar = 1

DIM T(1 TO N) AS SINGLE
DO
    FOR Count1 = 1 TO N
        PRINT
        PRINT "Ingrese la temperatura"; Count1; ", en °K";
        INPUT T(Count1)
    NEXT Count1
    PRINT
    PRINT
    PRINT "¿Los valores son correctos? (S, N)";
    INPUT Respuesta
    IF (Respuesta = "N") OR (Respuesta = "n") THEN
        Continuar = 0
    ELSE
        Continuar = 1
    END IF
LOOP UNTIL Continuar = 1

CLS
DO
    PRINT
    PRINT "Ingrese el número de Presiones";
    INPUT M
    PRINT
    PRINT
    PRINT "¿El valor es correcto? (S, N)";
    INPUT Respuesta
    IF (Respuesta = "N") OR (Respuesta = "n") THEN
        Continuar = 0
    ELSE
        Continuar = 1
    END IF
LOOP UNTIL Continuar = 1

DIM P(1 TO M) AS SINGLE
DO
    FOR Count1 = 1 TO M
        PRINT
        PRINT "Ingrese la Presión"; Count1; ", en atm.";
        INPUT P(Count1)
    NEXT Count1
    PRINT
    PRINT
    PRINT "¿Los valores son correctos? (S, N)";
    INPUT Respuesta
    IF (Respuesta = "N") OR (Respuesta = "n") THEN
        Continuar = 0
    ELSE
        Continuar = 1
    END IF
LOOP UNTIL Continuar = 1

CLS
DO
    PRINT
    PRINT "Ingrese la constante A0";
    INPUT A0
    PRINT
    PRINT "Ingrese la constante B0";
    INPUT B0
    PRINT
    PRINT "Ingrese la constante a";
    INPUT a
    PRINT
    PRINT "Ingrese la constante b";

```

```

INPUT b
PRINT
PRINT "Ingrese la constante c";
INPUT c
PRINT
PRINT "Ingrese la constante R, en (atm*lt)/(mol*K) ";
INPUT R
PRINT
PRINT
PRINT "¿Los valores son correctos? (S,N)";
INPUT Respuesta
IF (Respuesta = "N") OR (Respuesta = "n") THEN
  Continuar = 0
ELSE
  Continuar = 1
END IF
LOOP UNTIL Continuar = 1

CLS
DIM V(1 TO M) AS DOUBLE
DIM VC(1 TO N, 1 TO M) AS DOUBLE
DIM beta(1 TO N) AS DOUBLE
DIM delta(1 TO N) AS DOUBLE
DIM gamma(1 TO N) AS DOUBLE

FOR I = 1 TO N
  beta(I) = R*T(I)*B0 - A0 - R*c/T(I)^2
  gamma(I) = -R*T(I)*B0*b + A0*a - R*c*B0/T(I)^2
  delta(I) = R*B0*b*c/T(I)^2
  FOR J = 1 TO M
    V(J) = R*T(I)/P(J)
    DO
      función = R*T(I)/V(J) + beta(I)/V(J)^2 + gamma(I)/V(J)^3 + delta(I)/V(J)^4 - P(J)
      IF ABS(función) < .00001 THEN
        VC(I, J) = V(J)
        condición = 1
      ELSE
        derivada = -R*T(I)/V(J)^2 - 2*beta(I)/V(J)^3 - 3*gamma(I)/V(J)^4 - 4*delta(I)/V(J)^5
        V(J) = V(J) - función/derivada
        condición = 0
      END IF
    LOOP UNTIL condición = 1
  NEXT J
NEXT I

DIM Z(1 TO N, 1 TO M) AS DOUBLE
FOR I = 1 TO N
  FOR J = 1 TO M
    Z(I, J) = P(J)*VC(I, J)/(R*T(I))
  NEXT J
NEXT I

FOR I = 1 TO N
  PRINT
  PRINT "Para el valor de la temperatura:"; T(I); "°K"
  PRINT
  PRINT "El valor de Beta es:"; beta(I)
  PRINT
  PRINT "El valor de Gamma es:"; gamma(I)
  PRINT
  PRINT "El valor de Delta es:"; delta(I)
  SLEEP
  CLS
  FOR J = 1 TO M
    PRINT
    PRINT "El valor de la Presión "; J; ":"; P(J); "atm"
    PRINT
    PRINT "El valor del Volumen "; J; ":"; VC(I, J); "lts/mol"
    PRINT
    PRINT "El valor del Coeficiente de compresibilidad "; J; ":"; Z(I, J)
    SLEEP
  NEXT J
NEXT I
END

```